

Jürgen Enders

# Ein einfacher Plotter



Anwendungen mit TI-Innovator™ Hub und Rover

## Einleitung

Der TI-Innovator™ Hub mit TI Launchpad™ Board ist ein mit industriellen Komponenten aufgebautes Interface, das die Signale von Sensoren aufnehmen und Aktoren ansteuern kann. Dazu gibt es viele fertig aufgebaute Module, aber man kann auch eigene Schaltungen auf Steckplatinen (Breadboard) entwerfen und anschließen. Der TI-Innovator™ Hub funktioniert nur im Zusammenspiel mit einem TI-Nspire™CX / CAS, einem TI-Nspire™CX II-T / CAS oder einem TI-84 Plus CE-T bzw. der entsprechenden Computersoftware, da er auf die Stromversorgung dieser Geräte angewiesen ist. Auf diesen Geräten werden auch die Programme geschrieben, die für den Betrieb des TI-Innovator™ Hub notwendig sind. Die möglichen Programmiersprachen sind TI Basic oder LUA. Bei den folgenden Beispielen wird TI Basic verwendet.

## Kurze Einführung in das Programmieren mit TI Basic


Dies ist keine vollständige Einführung in die Programmiersprache, sondern anhand eines Beispiels eine Zusammenfassung einiger Befehle, die zum Verständnis der Beispielprogramme sinnvoll sind.

### AUFGABE:


Es soll ein Programm geschrieben werden, das folgendes leistet:

Nähert man sich dem Ultraschall-Entfernungssensor (Ranger) auf weniger als 10 cm, so ertönt dreimal eine kurze Warntonfolge und die RGB-LED leuchtet rot auf. Der Ranger wird mit dem Eingang IN 1 des Hub verbunden, der Taschenrechner mit dem USB-Port (Steckertyp B des kurzen Kabels). Das Programm soll durch Drücken der Taste `esc` beendet werden.


Soll ein neues Programm geschrieben werden, so fügt man eine neue Seite zu dem Dokument hinzu und wählt darin den Programmierer und die Auswahl `1:Neu ...`




Es öffnet sich ein Fenster, in dem ein Name für das Programm eingegeben werden muss. Im Beispiel wurde der Name `testd` gewählt.




Schließt man das Fenster, so kommt man in den Editiermodus. Alle Programmierbefehle werden zeilenweise in den Bereich zwischen *Prgm* und *EndPrgm* eingefügt, wo sich jetzt das gestrichelte Rechteck befindet. Jeder Befehl kommt in eine neue Zeile, Leerzeilen werden später bei der Ausführung des Programmes ignoriert.



Im *Menü* befinden sich in Gruppen zusammengefasst alle Programmierbefehle.




Das *Menü 4:Steuerung* enthält Befehle für Verzweigungen, Schleifen, usw.




Das *Menü 6:E/A* enthält alle Befehle zur Kommunikation mit dem Nutzer (Anzeigebefehle, Eingabebefehle)

Das *Menü 8:Hub* enthält alle Befehle zur Kommunikation mit dem TI-Innovator™ Hub.



Das *Menü 2:Syntax überprüfen* enthält Prüfbefehle, den Speicherbefehl sowie den Startbefehl für das Programm. Das Speichern bezieht sich allerdings nur auf den Arbeitsspeicher!

- a. Die Überschrift - der Befehl *DispAt* aus dem *Menü E/A* bewirkt, dass der in Anführungszeichen stehende Text *immer* in der 2. Zeile unter dem Trennstrich auf dem Home-Display dargestellt wird. *Disp* allein schreibt den Text immer in eine neue Zeile.
- b. Zur Verdeutlichung:  
 DispAt 1,... oben  
 DispAt 2,... unten



a

b

Der Befehl `Send „CONNECT RANGER 1 TO IN 1“` bewirkt die Zuordnung des RANGER 1 zum Eingang IN 1. Die Nummer bei RANGER gehört zwingend dazu und muss per Hand eingefügt werden.

Die Befehle finden sich im *Hub-Menü* unter `Send „CONNECT - Input, Settings und Ports.` Man kann den ganzen Text in den Anführungszeichen auch von der Tastatur eingeben, muss allerdings die Syntax genau (Großschreibung!) beachten. Der `Send`-Befehl bewirkt, dass die in den Anführungszeichen stehende Zeichenkette an den Hub gesendet wird.

Einfügen der zentralen *While*-Schleife:

`getKey()` liest den Tastaturcode einer gedrückten Taste. Solange man nicht die Taste `[esc]` gedrückt hat, werden die Befehle zwischen `While` und `EndWhile` ohne Ende wiederholt.

`While` findet man in *Steuerung, getKey* in E/A.

```

* testd 4/5
Define testd()=
Prgm
DispAt 2, "Nicht naeher als 10 cm"
Send "CONNECT RANGER 1 TO IN 1 "
While getKey()!="esc"
|
|
EndWhile
EndPrgm

```

Innerhalb der *While*-Schleife wird durch den Befehl `Send „READ RANGER 1“` aus dem Menü `Send „READ` fortlaufend die Entfernung gemessen (in der Maßeinheit m) und in einem Zwischenspeicher auf dem Hub abgelegt. Bei der nächsten Messung würde der Wert sofort überschrieben werden; deshalb muss er vorher durch den Befehl `Get` ausgelesen und einer Variablen zugewiesen werden, hier der Variablen `d` (von *D*istance; die Wahl des Variablennamens ist aber beliebig).

```

* testd 6/7
Define testd()=
Prgm
DispAt 2, "Nicht naeher als 10 cm"
Send "CONNECT RANGER 1 TO IN 1 "
While getKey()!="esc"
  Send "READ RANGER 1"
  Get d
|
|
EndWhile
EndPrgm

```

Einfügen der Verzweigung *If ... Then ... EndIf*:

Ist  $d < 0,1$  m, so sollen Aktionen erfolgen.

$d < 0.1$  ist (ebenso wie `getKey() != "esc"` in der *While*-Schleife) eine Bedingung, die entweder wahr (1) oder falsch (0) ist.

Ist sie wahr, so soll die eingebaute RGB-LED rot leuchten. Der Befehl `Send „SET COLOR.RED` befindet sich im Menü `Send „SET`. Die Einstellungen `ON` und `OFF` befinden sich im Menü `SETTINGS`. Man darf nicht vergessen, die LED wieder auszuschalten, denn sonst leuchtet sie immer weiter, egal was passiert, selbst wenn das Programm beendet ist und man weiter editiert!

```

* testd 9/11
Send "CONNECT RANGER 1 TO IN 1 "
While getKey()!="esc"
  Send "READ RANGER 1"
  Get d
  If d<0.1 Then
    Send "SET COLOR.RED ON "
  Send "SET COLOR.RED OFF |"
  EndIf
EndWhile

```

Es fehlt noch die Warntonfolge. Im Menü *Send* "SET" befindet sich unter der Bezeichnung *SOUND* der eingebaute kleine und recht leise Lautsprecher. Der Befehl muss noch vervollständigt werden durch die Frequenz des zu hörenden Tones in Hz.

Die in der Aufgabe geforderte Tonfolge besteht hier aus zwei Tönen mit den Frequenzen 440 Hz und 220 Hz, die beide 0,5 s lang ertönen sollen.

Dafür sorgt der Befehl *Wait*, der die weitere Ausführung des Programmes um die angegebene Zeit (hier 0,5 s) anhält.

Die Tonfolge wird dreimal innerhalb einer *For*-Schleife abgespielt:

*i* ist die Schleifenvariable (Name beliebig)

1 ist der Startwert, von dem aus in Schritten von 1 bis zum Endwert 3 hochgezählt wird.

Einmal eingeschaltet, würde auch der

Lautsprecher unbegrenzt weiter laufen; deshalb wird er mit dem Befehl *Send* "SET SOUND OFF" ausgeschaltet.

```

1.1 1.2 *testd 17/17
Send "SET COLOR.RED ON "
For i,1,3
  Send "SET SOUND 440"
  Wait 0.5
  Send "SET SOUND 220"
  Wait 0.5
EndFor
Send "SET COLOR.RED OFF "
Send "SET SOUND OFF "
EndIf

```

### Starten des Programmes:

Das geschieht in zwei Schritten:

1. Menü 2: 1: Syntax überprüfen & speichern
2. Menü 2: 3: Starten

Jetzt befindet man sich im *Calculate*-Bereich des Taschenrechners. Mit einem Druck auf  wird das Programm gestartet.

### Abbrechen eines Programmes:

Durch einen Fehler bei der Programmierung kann ein Programm endlos weiterlaufen. Man kann es jedoch unterbrechen


- auf dem Handheld durch Drücken von  und mehrfach

- auf dem PC durch *F12* und *Eingabe*.

### Fehlermeldungen:

links: bei der Syntaxüberprüfung (0,1 statt 0.1)

rechts: bei der Programmausführung; mit *Gehe zu* kommt man in den Editiermodus und in die Nähe des Fehlers.



Fehlermeldungen, die Befehle für den Hub betreffen, werden durch einen kurzen Piepton und das Aufblitzen der roten Fehler-LED auf dem Hub angezeigt; das Programm läuft aber weiter.

## Beispiele für den ROVER

### Ein einfacher Plotter

Der Rover ist ein Fahrzeug, das von zwei getrennt ansteuerbaren Motoren angetrieben wird. Es ist so gebaut, dass es auf der Stelle wenden kann. Fest eingebaut sind ein Front - Entfernungssensor (Ranger), ein Farbsensor zur Spurverfolgung sowie eine RGB-LED auf der Oberseite und ein Stifthalter zur Aufnahme von Filzstiften. Der Rover benötigt zu seinem Betrieb einen TI-Innovator™, der in ein Fach an der Unterseite so eingeschoben wird, dass die Ports IN1 - IN3 sowie OUT1 - OUT3 zugänglich bleiben. Über einen Breadboard-Port-Stecker werden Rover und Innovator miteinander verbunden. Ferner benötigt der Rover noch einen Taschenrechner ( TI-Nspire™CX (CAS) oder TI-84 Plus CE-T), der das Programm liefert und auf der Oberfläche des Rovers montiert werden kann. Ein Betrieb am PC ist nicht zu empfehlen, da er kabelgebunden erfolgen muss.

Der Rover enthält einen eigenen Akku, der etwa 6 1/2 h hält. In dieser Zeit legt der Rover ungefähr 3800 m mit der voreingestellten Standardgeschwindigkeit zurück.

Der Befehl *Send „RV TO XY x-Koordinate y-Koordinate“* veranlasst den Rover, zum Punkt P mit den angegebenen Koordinaten zu fahren. Die Position dieses Punktes ist dabei abhängig von der anfänglichen Orientierung des Rovers: der Punkt, in dem der Zeichenstift auf die Zeichenebene auftritt, ist der Nullpunkt und der Rover zeigt vorwärts in Richtung der positiven x-Achse. Das virtuelle Raster, auf dem sich der Rover bewegt, hat eine Kantenlänge von 0,1 m pro Einheit. Lässt man den Rover eine Folge von Punkten anfahren, so wird nur der erste Punkt vom virtuellen Nullpunkt aus angefahren, alle Folgepunkte werden aus der jeweils aktuellen Position erreicht.

#### Aufgabe:

Der Rover soll den Graphen einer Funktion zeichnen, die in der üblichen oder in der parametrischer Form vorgegeben ist.



#### Hinweise:

1. Die Funktionen müssen im jeweiligen Grafikfenster als  $f_1(x)$  bzw.  $x_1(t)$  und  $y_1(t)$  deklariert werden. Dabei gilt für die Fensterparameter  $-5 \leq x \leq 5$   $-5 \leq y \leq 5$  und  $0 \leq t \leq 2\pi$ .
2. Für die Spur sollte ein dünner Filzstift in den Rover eingesetzt und ein hinreichend großes Stück Papier untergelegt werden.

#### Programm:

Die beiden Bilder zeigen die im Beispiel verwendeten Funktionen.

Alle Punkte werden zunächst in einer Liste gespeichert. Der Rover fährt ausgehend vom Nullpunkt zum ersten Punkt des Graphen. Der Graph selbst ist dann ein Polygonzug, bestehend aus 25 Abschnitten.



Hier geht es zunächst nur um die Überschriften.

Durch den Befehl *Request* wird die Programmausführung angehalten.  
 „f“ steht für die üblichen funktionale Darstellung.  
 Es werden 26 x-Werte im Abstand von 0,4 LE berechnet und in der Liste *lx* abgespeichert.  
 Die y-Werte stehen in der Liste *ly*, wobei die *if*-Verzweigungen dafür sorgen, dass der Bereich von  $-5 \leq y \leq 5$  nicht überschritten wird.

„p“ (oder auch jeder andere Buchstabe) führt zur parametrischen Darstellung. Hier werden ebenfalls 26 Werte erzeugt, jetzt aber für den Parameter *t* im Bereich von  $0 \leq t \leq 2\pi$ .  
 Die Liste *lx* muss nun ebenfalls auf den Bereich  $-5 \leq x \leq 5$  beschränkt werden.

Erst jetzt wird der Rover verbunden.  
 Dann fährt er die Punkte  $P(lx[i]/ly[i])$  der Reihe nach an und hinterlässt als Spur einen Polygonzug.

```

Define plotter()=
Prgm
:DispAt 1,"Plotter"
:DispAt 2,"funktional oder
parametrisch"
:DispAt 3,"Definition aus dem
Grafikfenster"
:DispAt 4,"-5≤x≤5   -5≤y≤5
0≤t≤2π"
:RequestStr "f/p ?",w
:If w="f" Then
:   For i,1,26
:     lx[i]:=-5.4+i*0.4
:     ly[i]:=f1(lx[i])
:     If ly[i]≤-5
:       ly[i]:=-5
:     If ly[i]≥5
:       ly[i]:=5
:   EndFor
:Else
:   For i,1,26
:     t:=((i-1)*2*π)/(25)
:     lx[i]:=x1(t)
:     If lx[i]≤-5
:       lx[i]:=-5
:     If lx[i]≥5
:       lx[i]:=5
:     ly[i]:=y1(t)
:     If ly[i]≤-5
:       ly[i]:=-5
:     If ly[i]≥5
:       ly[i]:=5
:   EndFor
:EndIf
:Send "CONNECT RV"
:For i,1,26
:  Send "RV TO XY eval(lx[i])
eval(ly[i])"
:EndFor
:EndPrgm

```

Haben Sie Fragen zu Produkten von Texas Instruments? Oder sind Sie an weiteren Unterrichtsmaterialien oder einer Lehrerfortbildung interessiert? Gerne steht Ihnen auch unser Customer Service Center mit Rat und Tat zu Seite. Nehmen Sie mit uns Kontakt auf:



Customer Service Center  
TEXAS INSTRUMENTS  
[education.ti.com/csc](http://education.ti.com/csc)

[education.ti.com/deutschland](http://education.ti.com/deutschland)  
[education.ti.com/oesterreich](http://education.ti.com/oesterreich)  
[education.ti.com/schweiz](http://education.ti.com/schweiz)

Weitere Materialien finden Sie unter:  
[www.ti-unterrichtsmaterialien.net](http://www.ti-unterrichtsmaterialien.net)

Dieses und weiteres Material steht Ihnen auf der TI Materialdatenbank zum Download bereit:  
[www.ti-unterrichtsmaterialien.net](http://www.ti-unterrichtsmaterialien.net)

© 2020 Texas Instruments

Dieses Werk wurde in der Absicht erarbeitet, Lehrerinnen und Lehrern geeignete Materialien für den Unterricht an die Hand zu geben. Die Anfertigung einer notwendigen Anzahl von Fotokopien für den Einsatz in der Klasse, einer Lehrerfortbildung oder einem Seminar ist daher gestattet. Hierbei ist auf das Copyright von Texas Instruments hinzuweisen. Jede Verwertung in anderen als den genannten oder den gesetzlich zugelassenen Fällen ist ohne schriftliche Genehmigung von Texas Instruments nicht zulässig. Alle Warenzeichen sind Eigentum ihrer Inhaber.